
Ausgabe 13/2011 

Software-Anleitung 

Projekte mit Excel 2007/2010 steuern 
Teil 4: Projektübersicht mit dynamisch angepasster Zeitleiste 
In den ersten drei Teilen haben Sie erfahren, wie Sie Projekte in Excel 
strukturieren, Termine planen, die Projektdauer, Projekttermine sowie 
Meilensteine berechnen und die Projektplanung mit Hilfe eines Gantt-
Diagramms übersichtlich darstellen. Dieser vierte und letzte Teil zeigt, 
wie Sie die Projektplanung auch mit einer Zeitleiste und Balken darstellen 
können, die sich dynamisch an neu eingetragene Daten anpassen. Diese 
Variante ist zwar aufwändiger zu erstellen als das einfache Gantt-
Diagramm, doch es entfällt das manuelle Anpassen der Zeitachse an den 
jeweiligen Zeitraum. Zudem bietet die dynamische Lösung mehr Flexibili-
tät hinsichtlich der optischen Darstellung. 

Gantt-Diagramm mit dynamischer Zeitleiste  
Bild 1 zeigt die zugrundeliegenden Daten und das daraus erzeugte 
Gantt-Diagramm mit dynamischer Zeitleiste.  

•  In die Spalten A bis D werden die Projektdaten, also die jeweiligen 
Arbeitspakte und Meilensteine sowie deren Anfangs- und Enddatum 
eingegeben.  

• In den Spalten E und F errechnet Excel die Differenz zwischen An-
fangs- und Enddatum und zwar einmal in Kalendertagen (KT) und einmal in tatsächlich verfügbaren Arbeits-
tagen (PT = Projekttage). 

• Ab Spalte G stellt Excel mit Hilfe automatisch erzeugter, farbiger Balken dar, wann die einzelnen Arbeitspakete 
beginnen und enden. 

Bild 1:  Vorschau auf die fertige Lösung, die mit Formeln sowie mit Bedingter Formatierung arbeitet. 

• Die zum Balkendiagramm gehörigen Datums-Informationen in den Zeilen 2 und 3 erzeugt Excel ebenfalls  
automatisch – und zwar in Abhängigkeit von den in den Spalten C und D eingetragenen Beginn- und Endedaten. 
Das Diagramm "wächst" so beim Eingeben der Daten automatisch nach rechts (im Beispiel bis Spalte NZ). 

• Wochenenden werden in Zeile 2 und 3 mit hellgrauer Zellfarbe kenntlich gemacht. 

• Durch Vergabe einer Kategorie in Spalte B lassen sich die Pakete in der Balkendarstellung farblich differenzieren 
– beispielsweise um Projekte, Prioritäten, Themen oder Abteilungen zu unterscheiden. Im Beispiel können die 
Kategorien A, B oder C vergeben werden.  

• Meilensteine werden mit einem M gekennzeichnet. Sie werden in der Balkendarstellung in einem kräftigen 
Orange dargestellt. 

Diese Lösung eignet sich nicht nur für die Übersicht in einem Projekt, sondern damit lassen sich beispielsweise 
auch verschiedenen Projekte innerhalb eines Kalenderjahres abbilden. 

Die Anforderungen an die dynamische Lösung 

• Zu jedem der Arbeitspakete ist die Eingabe von vier Daten in den Spalten A bis D erforderlich sein, nämlich: 
Start und Ende, Kategorie (A, B, C bzw. M) und natürlich die Bezeichnung. 

• Beginn und Ende der dynamischen Zeitleiste in Zeile 2 orientieren sich am niedrigsten Datumswert in Spalte C 
und am höchsten Datumswert in Spalte D. 

• Auf Basis der Datumswerte in Zeile 2 werden in Zeile 3 die zugehörigen Wochentage (abgekürzt) angezeigt. 

• Wenn in Spalte C und D Datumswerte eingetragen wurden, generieren sich ab Spalte G die dazu passenden 
Balken mit Hilfe der Bedingten Formatierung. 

• Die farbliche Unterscheidung der Balken bzw. der Meilensteine ergibt sich aus den Werten A, B, C oder M in 
Spalte B. Hier ließe sich mittels Gültigkeitsprüfung noch sicherstellen, dass tatsächlich nur diese vier Werte 
eingetragen werden können. Im Beispiel wurde auf eine solche Gültigkeitsprüfung verzichtet. Falls Sie diese 
ergänzen möchten, können Sie sich an der Beschreibung im ersten Teil dieser Artikelfolge orientieren.  

• In der Lösung wird davon ausgegangen, dass Daten für die einzelnen Arbeitspakete ab Zeile 5 eingegeben werden. 

Kalender- und tatsächliche Projekttage berechnen 
Die Berechnung von Kalender- und Arbeitstagen (Projekttagen) in den Spalten E und F erfolgt mit Hilfe der For-
meln und Funktionen, die Sie bereits in den vergangenen Teilen kennengelernt haben. Allerdings müssen auch 
hier die Berechnungen an bestimmte Bedingungen geknüpft werden. Sie dürfen erst erfolgen, wenn in Spalte C 
und D Datumswerte für Beginn und Ende eines Pakets eingegeben wurden. 

Die Formel zur Berechnung der Kalendertage 

Die Formel =D5-C5+1 zur Berechnung der Kalendertage in Zelle E5 muss demzufolge in eine WENN-Funktion 
eingebaut werden, die prüft, ob in C5 und in D5 ein Datumswert steht. Das Vorhandensein von Datumswerten 

ermitteln Sie mit der Funktion ISTZAHL. Da in Spalte C und in Spalte D ein Datum stehen muss, bevor Sie die 
Anzahl der Kalendertage berechnen können, werden die beiden ISTZAHL-Funktionen mit einem UND verknüpft. 
Die vollständige Formel für Zelle E5 lautet also: 

=WENN(UND(ISTZAHL(C5);ISTZAHL(D5));D5-C5+1;0) 

Die Formel zur Berechnung der tatsächlichen Projekttage 

Nach dem gleichen Prinzip ist in Spalte F auch die Formel zum Berechnen der Projekttage aufgebaut, bei der die 
Funktion NETTOARBEITSTAGE zum Einsatz kommt. Auch hier muss zuerst geprüft werden, ob in C und D Da-
tumswerte eingetragen wurden. Die vollständige Formel für Zelle F5 lautet demzufolge: 

=WENN(UND(ISTZAHL(C5);ISTZAHL(D5));NETTOARBEITSTAGE(C5;D5;Freie_Tage);0) 

Die Formel nach unten ausfüllen 

Um die beiden Formeln in die darunterliegenden Zellen zu kopieren, markieren Sie zunächst den Zellbereich 
E5:F5 und ziehen Sie dann das Ausfüllkästchen nach unten bis Zeile 22. Alternativ dazu können Sie auch den 
Bereich E5 bis F22 markieren und dann die beiden Formeln mit der Tastenkombination Strg+U nach unten kopieren. 

Beginn und Ende der Zeitleiste dynamisch gestalten 
Die Zeitleiste soll in der zweiten Zeile ab der Spalte G beginnen. Wenn in C5 das erste Startdatum eingetragen 
wird, soll dies den Beginn der Zeitleiste in G2 bestimmen. Allerdings reicht dafür die einfache Formel =C5 nicht aus.  

• Vorher ist noch zu prüfen, ob in C5 bereits ein Datum eingetragen wurde. Das erledigen Sie erneut mit der 
Funktion ISTZAHL. 

• Um auch tatsächlich das früheste Startdatum, also den kleinsten Wert aus Spalte C für den Beginn der Zeit-
leiste zu ermitteln, verwenden Sie die Funktion MIN, die aus einem Zellbereich – hier C5 bis C22 – den Mini-
malwert herausfindet. Damit ergibt sich für G2 die folgende Formel: 

=WENN(ISTZAHL(C5);MIN(C5:C22);"") 

Folgetage und das Ende der Zeitleiste bestimmen 

Bis zu welcher Spalte soll die Zeitleiste ab Spalte H ergänzt werden? Die Antwort auf diese Frage ermitteln Sie mit 
einer etwas komplexeren Formel. Diese muss drei Bedingungen prüfen: 

• Analog zur Formel für G2 muss zunächst ermittelt werden: Steht in C5 bereits ein erstes Datum? 

• Außerdem ist zu prüfen, ob der Datumswert in der benachbarten linken Zelle größer oder gleich dem kleinsten 
Datum in Spalte C ist. 

• Schließlich ist noch zu klären, ob der Datumswert in der benachbarten linken Zelle kleiner ist als der größte 
Endwert in Spalte D. 

Sind alle drei Bedingungen erfüllt, kann es ab Spalte H ein Folge- oder Enddatum in der Zeitleiste geben. Excel 
soll dann also das Datum um einen Zähler erhöhen (Datum der Vorgängerzelle plus 1). Anderenfalls soll die Zelle 
leer bleiben. Das erreichen Sie mit einer Anweisung, die einfach aus zwei Anführungszeichen besteht. Die einzel-
nen Bedingungen müssen auch hier wieder mit UND verknüpft werden. Die vollständige Formel für H2 lautet damit: 

=WENN(UND(ISTZAHL($C$5);G2>=MIN($C$5:$C$22);G2<MAX($D$5:$D$22));G2+1;"") 

Diese Formel können Sie durch Ziehen am Ausfüllkästchen nach rechts kopieren. Im Beispiel wurde die Formel 
bis zur Spalte NZ übertragen – das deckt den Zeitraum eines Kalenderjahrs ab. 

Die Information zum jeweiligen Wochentag ergänzen 
In Zeile 3 soll nun zu jedem Datum, das in Zeile 2 steht, zusätzlich der entsprechende Wochentag angezeigt  
werden. Bevor Sie diese Zusatzinformation einblenden, muss geprüft werden, ob in der gleichen Spalte in Zeile 2 
schon ein Eintrag steht. Gehen Sie wie folgt vor: 

• Markieren Sie Zelle G3. 

• Geben Sie die folgende Formel 
ein:=WENN(ISTZAHL(G2);G2;""). 

• Rufen Sie mit Strg+1 das Dia-
logfeld zum Formatieren von 
Zellen auf. Klicken Sie auf der 
Registerkarte "Zahlen" links un-
ten die Kategorie "Benutzerdefi-
niert" an. 

• Tragen Sie rechts im Eingabe-
feld unter "Typ" den Formatcode 
"TTT" ein. Damit wird das Da-
tum aus Zeile 2 auf die Anzeige 
des abgekürzten Wochentags 
reduziert. 

• Kopieren Sie diese Formel und das eben definierte Zahlenformat mit dem Ausfüllkästchen nach rechts bis zur 
Spalte NZ. 

Bild 2: Das benutzerdefinierte Zahlenformat für die Anzeige des abgekürzten Wochen-
tags anlegen. 

Wochenenden in der Zeitleiste farbig kennzeichnen 
Damit die Betrachter der Übersicht schnell erkennen, wann Wochenende ist, sorgen Sie mittels Bedingter Forma-
tierung dafür, dass alle Datumsangaben, die auf einen Samstag oder Sonntag fallen, mit der Farbe Hellgrau hin-
terlegt werden. Gehen Sie dazu wie folgt vor: 

• Markieren Sie den Zellbereich 
von G2 bis NZ3. 

• Klicken Sie in der Registerkarte 
Start auf die Schaltfläche Be-
dingte Formatierung. 

• Wählen Sie "Neue Regel ..." 
und dann "Formel zur Ermittlung 
der zu formatierenden Zellen 
verwenden". 

• Tragen Sie – wie in Bild 3 ge-
zeigt – die folgende Formel ein: 
=UND(ISTZAHL(G$2);WOCHE
NTAG(G$2;2)>5). Diese Formel 
prüft, ob in Zeile 2 der gleichen 
Spalte ein Datum steht und ob 
der Wochentag für dieses Da-
tum nach dem Freitag (Zahl 5) 
liegt. Sind beide Bedingungen 
erfüllt, nimmt die Formel den 
Wert "wahr" an. 

• Klicken Sie nach Eingabe dieser Formel rechts auf die Schaltfläche "Formatieren". 

• Wählen Sie im folgenden Dialogfeld in der Registerkarte "Ausfüllen" einen passenden hellgrauen Farbton. 

• Schließen Sie den Vorgang mit zweimaligem Klick auf "OK" ab. 

Beginn, Länge und Farbe der Balken definieren 
Die Lösung ist so aufgebaut, dass in den Zeilen mit ungerader Nummer die Informationen zu den Arbeitspaketen, 
Phasen oder Projekten eingegeben und anschließend ab Spalte G dargestellt werden. Die Zeilen mit geraden 
Nummern dienen als Leerräume zwischen den Balken. 

Die Balken ab Spalte G entstehen durch Einfärben des Zellhintergrunds. Allerdings erfolgt das Zuweisen der Hin-
tergrundfarbe nicht manuell, sondern automatisch mit Hilfe der Bedingten Formatierung. Damit dieser Automatis-
mus funktioniert, sind vorab folgende Informationen zu prüfen: 

Bild 3: Die Formatierungsregel, die für eine hellgraue Hinterlegung von Wochenenden sorgt. 

• Handelt es sich bei der Zeile um eine mit einer ungeraden Nummer? Das erledigt die Funktion UNGERADE  
in Kombination mit der Funktion ZEILE. 

• Wurde in der Zeile schon ein Start- und Endedatum eingegeben – stehen also in C5 und D5 Datumsangaben? 
Das wird wieder mit ISTZAHL überprüft. 

• Außerdem ist zu ermitteln, ob in der aktuellen Spalte das Datum aus Zeile 2 dem Startdatum in Spalte C  
entspricht oder größer ist und vor oder auf dem Endedatum liegt. 

• Zum Schluss wird noch ermittelt, welcher Buchstabe in Spalte B steht, d.h. welcher Kategorie der Vorgang 
zugewiesen wurde. Denn aus dieser Information resultiert die Farbe des Balkens bzw. des Meilensteins. 

Sind die Voraussetzungen geklärt, kann das erste bedingte Format definiert werden. 

• Markieren Sie den Bereich G5 bis NZ22. 

• Klicken Sie auf die Schaltfläche Bedingte Formatierung. 

• Wählen Sie "Neue Regel" und dann "Formel zur Ermittlung der zu formatierenden Zellen verwenden". 

• Tragen Sie dann die folgende Formel ein: 
=UND(ISTUNGERADE(ZEILE());ISTZAHL($C5);ISTZAHL($D5);G$2>=$C5;G$2<=$D5;$B5="A"). 

• Klicken Sie auf die Schaltfläche "Formatieren" und wählen Sie in der Registerkarte "Ausfüllen" eine  
passende Farbe. 

• Schließen Sie mit "OK" ab. 

Definieren Sie nun auf die gleiche 
Weise für die Arbeitspakte der Katego-
rie B und C sowie für alle Meilensteine 
(Kategorie = M) drei weitere Regeln. 
Stellen Sie dabei jeweils bei "Ausfül-
len" einen anderen Farbton ein. Da 
Excel ab Version 2007 beliebig viele 
bedingte Formate für eine Zelle zu-
lässt, können Sie außer A, B und C bei 
Bedarf auch weitere Farbkategorien 
anlegen. In älteren Excel-Versionen ist 
die Anzahl der Bedingten Formatierungen dagegen auf drei limitiert. Die entsprechenden Formeln für die Kategorien 
B und C sowie für Meilensteine lauten: 

=UND(ISTUNGERADE(ZEILE());ISTZAHL($C5);ISTZAHL($D5);G$2>=$C5;G$2<=$D5;$B5="B") 
=UND(ISTUNGERADE(ZEILE());ISTZAHL($C5);ISTZAHL($D5);G$2>=$C5;G$2<=$D5;$B5="C") 
=UND(ISTUNGERADE(ZEILE());ISTZAHL($C5);ISTZAHL($D5);G$2>=$C5;G$2<=$D5;$B5="M") 

Bild 4:  Die vier Regeln, die Position, Länge und Farbe der Balken bestimmen. 

Fazit 
Mit der hier vorgestellten Lösung verfügen Sie über ein leistungsfähiges Werkzeug, um Termine und Planungen in 
der Projektarbeit zu visualisieren. Die Möglichkeiten der Bedingten Formatierung werden durch den Einsatz von 
Formeln und Funktionen deutlich erweitert. Die Tatsache, dass in Excel 2007 und 2010 auch mehr als nur drei 
Kriterien bei der Bedingte Formatierung zur Verfügung stehen, macht es möglich, auch komplexe Sachverhalte zu 
handhaben. 

 

 

Hat Ihnen dieser Artikel gefallen? 

Bewerten Sie ihn im Projekt Magazin online und teilen Sie so Ihre Meinung anderen Lesern mit. Wählen Sie 
dazu den Artikel im Internet unter www.projektmagazin.de/ausgaben/2011 oder klicken Sie hier, um direkt 
zum Artikel zu gelangen. 

